

# xmerl Application

version 1.0

Typeset in L<sup>A</sup>T<sub>E</sub>X from SGML source using the DOCBUILDER 3.3.2 Document System.

# Contents

<b>1</b>	<b>xmerl User's Guide</b>	<b>1</b>
1.1	xmerl . . . . .	1
1.1.1	Introduction . . . . .	1
1.1.2	xmerl User Interface Data Structure . . . . .	2
1.1.3	Getting Started . . . . .	3
1.1.4	Example: Extracting Data From XML Content . . . . .	4
1.1.5	Example: Create XML Out Of Arbitrary Data . . . . .	5
1.1.6	Example: Transforming XML To HTML . . . . .	7
1.2	xmerl Release Notes . . . . .	9
1.2.1	xmerl 1.0 . . . . .	9
<b>2</b>	<b>xmerl Reference Manual</b>	<b>11</b>
2.1	xmerl . . . . .	14
2.2	xmerl_eventp . . . . .	17
2.3	xmerl_scan . . . . .	18
2.4	xmerl_xpath . . . . .	21
2.5	xmerl_xs . . . . .	23
	<b>Index of Modules and Functions</b>	<b>25</b>



# Chapter 1

## xmerl User's Guide

The *xmerl* application contains modules with support for processing of xml files compliant to XML 1.0.

### 1.1 xmerl

#### 1.1.1 Introduction

##### Features

The *xmerl* XML parser is able to parse XML documents according to the XML 1.0 standard. As default it performs well-formed parsing (syntax checks and checks of well-formed constraints). Optionally one can also use *xmerl* as a validating parser (validate according to referenced DTD and validating constraints). By means of for example the *xmerl\_xs* module it is possible to transform the parsed result to other formats, e.g. text, HTML, XML etc.

##### Overview

This document does not give an introduction to XML. There are a lot of books available that describe XML from different views. At the [www.W3.org](http://www.W3.org)<sup>1</sup> site you will find the XML 1.0 specification<sup>2</sup> and other related specs. One site where you can find tutorials on XML and related specs is [ZVON.org](http://ZVON.org)<sup>3</sup>.

However, here you will find some examples of how to use *xmerl* and some scenarios in which it can be used. A detailed description of the user interface can be found in the reference manual<sup>4</sup>.

There are two known shortcomings in *xmerl*:

- It cannot retrieve external entities on the Internet by a URL reference, only resources in the local file system.
- *xmerl* can parse Unicode encoded data. But, it fails on tag names, attribute names and other markup names that are encoded Unicode characters not mapping on ASCII.

---

<sup>1</sup>URL: <http://www.w3.org>

<sup>2</sup>URL: <http://www.w3.org/TR/REC-xml/>

<sup>3</sup>URL: <http://www.zvon.org>

<sup>4</sup>URL: [application\\_frame.html](#)

By parsing an XML document you will get a record, displaying the structure of the document, as return value. The record also holds the data of the document. xmerl is convenient to use in for instance the following scenarios:

You need to retrieve data from XML documents. Your Erlang software can handle information from the XML document by extracting data from the data structure received by parsing.

It is also possible to do further processing of parsed XML with xmerl. If you want to change format of the XML document to for instance HTML, text or other XML format you can transform it. There is support for such transformations in xmerl.

One may also convert arbitrary data to XML. Then it is easy to for instance make it readable by humans. In this case you first create xmerl data structures out of your data, then transform it to XML.

You can find usage examples of these three scenarios below.

## 1.1.2 xmerl User Interface Data Structure

The following records used by xmerl to save the parsed data are defined in `xmerl.hrl`

The result of a successful parsing is a tuple `{DataStructure,M}`. `M` is the XML production Misc, which is the markup that comes after the element of the document. It is returned "as is". `DataStructure` is an `xmlElement` record, that among others have the fields `name`, `parents`, `attributes` and `content` like:

```
#xmlElement{name=Name,  
            ...  
            parents=Parents,  
            ...  
            attributes=Attrs,  
            content=Content,  
            ...}
```

The name of the element is found in the `name` field. In the `parents` field are the names of the parent elements saved. `Parents` is a list of tuples where the first element in each tuple is the name of the parent element. The list is in reverse order.

The record `xmlAttribute` holds the name and value of an attribute in the fields `name` and `value`. All attributes of an element is a list of `xmlAttribute` in the field `attributes` of the `xmlElement` record.

The `content` field of the top element is a list of records that shows the structure and data of the document. If it is a simple document like:

```
<?xml version="1.0"?>  
<dog>  
Grand Danois  
</dog>
```

The parse result will be:

```
#xmlElement{name = dog,  
            ...  
            parents = [],  
            ...  
            attributes = [],  
            content = [{xmlText, [{dog,1}],1,[], "\nGrand Danois\n",text}],  
            ...  
            }
```

Where the content of the top element is: `[{xmlText, [{dog,1}],1, [], "\nGrand Danois\n",text}]`. Text will be returned in `xmlText` records. Though, usually documents are more complex, and the content of the top element will in that case be a nested structure with `xmlElement` records that in turn may have complex content. All of this reflects the structure of the XML document.

Space charactes between markup as `space`, `tab` and `line feed` are normalized and returned as `xmlText` records.

## Errors

An unsuccessful parse results in an error, which may be a tuple `{error, Reason}` or an exit: `{'EXIT', Reason}`. According to the XML 1.0 standard there are `fatal error` and `error situations`. The fatal errors *must* be detected by a conforming parser while an error *may* be detected. Both categories of errors are reported as fatal errors by this version of `xmerl`, most often as an exit.

## 1.1.3 Getting Started

If you want to parse an XML file you run it in the Erlang shell like:

```
3> {ParsResult,Misc}=xmerl_scan:file("people.xml"5).
{{xmlElement,people,
  people,
  [],
  {xmlNamespace,[],[]},
  [],
  1,
  [],
  [{xmlText,[{people,1}],1,[], "\n  ",text},
   {xmlElement,person,
    person,
    [],
    {xmlNamespace,[],[]},
    [{people,1}],
    2,
    [{xmlAttribute,born,[],[],[],[]|...},
     {xmlAttribute,died,[],[],[]|...}],
    [{xmlText,[{person,2},{people|...}],
      1,
      []|...},
     {xmlElement,name,name,[],[]|...},
     {xmlText,[{...}|...],3|...},
     {xmlElement,profession|...},
     {xmlText|...},
     {...}|...}],
    [],
    ". ",
    undeclared},
   {xmlText,[{people,1}],3,[], "\n  ",text},
   {xmlElement,person,
    person,
    [],
```

---

<sup>5</sup>URL: `people.txt`

```

        {xmlNamespace, [], []},
        [{people, 1}],
        4,
        [{xmlAttribute, born, [], [] | ...},
         {xmlAttribute, died, [] | ...}],
        [{xmlText, [{...} | ...], 1 | ...},
         {xmlElement, name | ...},
         {xmlText | ...},
         {...} | ...],
        [],
        ". ",
        undeclared},
    {xmlText, [{people, 1}], 5, [], "\n", text}],
    [],
    ". ",
    undeclared},
    "\n"}
4>

```

If you instead receive the XML doc as a string you can parse it by `xmerl_scan:string/1`. Both `file/2` and `string/2` exists where the second argument is a list of options to the parser, see the reference manual<sup>6</sup>.

#### 1.1.4 Example: Extracting Data From XML Content

In this and the following examples we use the XML file "motorcycles.xml"<sup>7</sup> and the corresponding DTD "motorcycles.dtd"<sup>8</sup>.

In this example consider the situation where you want to examine a particular data in the XML file. For instance you want to check for how long each motorcycle has been recorded.

Take a look at the DTD and observe that the structure of an XML document that is conformant to this DTD must have one motorcycles element (the root element). The motorcycles element must have at least one bike element. After each bike element it may be a date element. The content of the date element is #PCDATA (Parsed Character DATA), i.e. raw text. Observe that if #PCDATA must have a "<" or a "&" character it must be written as "&lt;" and "&amp;" respectively. Also other character entities exist similar to the ones in HTML and SGML.

If you successfully parse the XML file with the validation on, as in: `xmerl_scan:file('motorcycles.xml', [{validation, true}])` you know that the XML document is valid and has the structure according to the DTD.

Thus, knowing the allowed structure, it is easy to write a program that traverses the data structure and picks the information in the xmlElements records with name date.

Observe that white space, each space, tab or line feed, between markup results in an xmlText record.

<sup>6</sup>URL: [xmerl\\_scan.html](#)

<sup>7</sup>URL: [motorcycles.txt](#)

<sup>8</sup>URL: [motorcycles\\_dtd.txt](#)

### 1.1.5 Example: Create XML Out Of Arbitrary Data

For this task there is more than one way to go. The “brute force” method is to create the records you need and feed your data in the content and attribute fields of the appropriate element.

There is support for this in xmerl by the “simple-form” format. You can put your data in a simple-form data structure and feed it into `xmerl:export_simple(Content, Callback, RootAttributes)`. Content may be a mixture of simple-form and xmerl records as `xmlElement` and `xmlText`.

The Types are:

- Content = [Element]
- Callback = atom()
- RootAttributes = [Attributes]

Element is any of:

- {Tag, Attributes, Content}
- {Tag, Content}
- Tag
- IOString
- #xmlText{}
- #xmlElement{}
- #xmlPI{}
- #xmlComment{}
- #xmlDecl{}

The simple-form structure is any of {Tag, Attributes, Content}, {Tag, Content} or Tag where:

- Tag = atom()
- Attributes = [{Name, Value} | #xmlAttribute{}]
- Name = atom()
- Value = IOString | atom() | integer()

See also reference manual for xmerl<sup>9</sup>

If you want to add the information about a black Harley Davidsson 1200 cc Sportster motorcycle from 2003 that is in shape as new in the motorcycles.xml document you can put the data in a simple-form data structure like:

```
Data =
  {bike,
   [{year, "2003"}, {color, "black"}, {condition, "new"}],
   [{name,
     [{manufacturer, ["Harley Davidsson"]},
      {brandName, ["XL1200C"]},
      {additionalName, ["Sportster"]}]}],
   {engine,
    ["V-engine, 2-cylinders, 1200 cc"]},
   {kind, ["custom"]},
   {drive, ["belt"]}]}
```

<sup>9</sup>URL: [xmerl.html#export\\_simple-3](http://xmerl.html#export_simple-3)

In order to append this data to the end of the motorcycles.xml document you have to parse the file and add Data to the end of the root element content.

```
{RootEl,Misc}=xmerl_scan:file('motorcycles.xml'),
#xmlElement{content=Content} = RootEl,
NewContent=Content++lists:flatten([Data]),
NewRootEl=RootEl#xmlElement{content=NewContent},
```

Then you can run it through the `export_simple/2` function:

```
{ok,IOF}=file:open('new_motorcycles.xml',[write]),
Export=xmerl:export_simple([NewRootEl],xmerl_xml),
io:format(IOF,"~s~n",[lists:flatten(Export)]),
```

The result would be `new_motorcycles.xml`<sup>10</sup>. If it is important to get similar indentation and newlines as in the original document you have to add `#xmlText{}` records with space and newline values in appropriate places. It may also be necessary to keep the original prolog where the DTD is referenced. If so, it is possible to pass a `RootAttribute{prolog,Value}` to `export_simple/3`. The following example code fixes those changes in the previous example:

```
Data =
  [#xmlText{value="  "},
   {bike,[{year,"2003"},{color,"black"},{condition,"new"}],
    [#xmlText{value="\n  "},
     {name,[#xmlText{value="\n          "},
            {manufacturer,["Harley Davidsson"}],
            #xmlText{value="\n          "},
            {brandName,["XL1200C"}],
            #xmlText{value="\n          "},
            {additionalName,["Sportster"}],
            #xmlText{value="\n          "}}],
     {engine,["V-engine, 2-cylinders, 1200 cc"}],
     #xmlText{value="\n          "},
     {kind,["custom"}],
     #xmlText{value="\n          "},
     {drive,["belt"}],
     #xmlText{value="\n          "}}],
   #xmlText{value="\n"}],
  ...
NewContent=Content++lists:flatten([Data]),
NewRootEl=RootEl#xmlElement{content=NewContent},
...
Prolog = ["<?xml version=\"1.0\" encoding=\"utf-8\" ?>
<!DOCTYPE motorcycles SYSTEM \"motorcycles.dtd\">\n"],
Export=xmerl:export_simple([NewRootEl],xmerl_xml,[{prolog,Prolog}]),
...
```

The result will be `new_motorcycles2.xml`<sup>11</sup>.

---

<sup>10</sup>URL: `new_motorcycles.txt`

<sup>11</sup>URL: `new_motorcycles2.txt`

## 1.1.6 Example: Transforming XML To HTML

Assume that you want to transform the `motorcycles.xml`<sup>12</sup> document to HTML. If you want the same structure and tags of the resulting HTML document as of the XML document then you can use the `xmerl:export/2` function. The following:

```
2> {Doc,Misc}=xmerl_scan:file('motorcycles.xml').
{{xmlElement,motorcycles,
  motorcycles,
  [],
  {xmlNamespace,[],[]},
  [],
  1,
  [],
  [{xmlText,[{motorcycles,1}],1,[],"\n ",text},
  {xmlElement,bike,
...
3> DocHtml=xmerl:export([Doc],xmerl_html).
["<!DOCTYPE HTML PUBLIC \"",
"-//W3C//DTD HTML 4.01 Transitional//EN",
"\",
[],
">\n",
[[["<","motorcycles",">"],
  ["\n ",
  [["<","
  "bike",
  [[" ", "year", "=\n", "2000", "\n"], [" ", "color", "=\n", "black", "\n"]],
  ">"],
...

```

Will give the result `result_export.html`<sup>13</sup>

Perhaps you want to do something more arranged for human reading. Suppose that you want to list all different brands in the beginning with links to each group of motorcycles. You also want all motorcycles sorted by brand, then some flashy colours on top of it. Thus you rearrange the order of the elements and put in arbitrary HTML tags. This is possible to do by means of the XSL Transformation (XSLT)<sup>14</sup> like functionality in `xmerl`.

Even though the following example shows one way to transform data from XML to HTML it also applies to transformations to other formats.

`xmerl_xs` does not implement the entire XSLT specification but the basic functionality. For all details see the reference manual<sup>15</sup>

First, some words about the `xmerl_xs` functionality:

You need to write template functions to be able to control what kind of output you want. Thus if you want to encapsulate a bike element in `<p>` tags you simply write a function:

```
template(E = #xmlElement{name='bike'}) ->
  ["<p>",xslapply(fun template/1,E),"</p>"];

```

<sup>12</sup>URL: `motorcycles.txt`

<sup>13</sup>URL: `result_export.html`

<sup>14</sup>URL: <http://www.w3.org/Style/XSL/>

<sup>15</sup>URL: `xmerl_xs.html`

With `xslapply` you tell the XSLT processor in which order it should traverse the XML structure. By default it goes in preorder traversal, but with the following we make a deliberate choice to break that order:

```
template(E = #xmlElement{name='bike'}) ->
  ["<p>",xslapply(fun template/1,select("bike/name/manufacturer")), "</p>"];
```

If you want to output the content of an XML element or an attribute you will get the value as a string by the `value_of` function:

```
template(E = #xmlElement{name='motorcycles'}) ->
  ["<p>",value_of(select("bike/name/manufacturer",E), "</p>");
```

In the `xmerl_xs` functions you can provide a `select(String)` call, which is an XPath<sup>16</sup> functionality. For more details see the `xmerl_xs` tutorial<sup>17</sup>.

Now, back to the example where we wanted to make the output more arranged. With the template:

```
template(E = #xmlElement{name='motorcycles'}) ->
  [
    "<head>\n<title>motorcycles</title>\n</head>\n",
    "<body>\n",
    "<h1>Used Motorcycles</h1>\n",
    "<ul>\n",
    remove_duplicates(value_of(select("bike/name/manufacturer",E))),
    "\n</ul>\n",
    sort_by_manufacturer(xslapply(fun template/1, E)),
    "</body>\n",
    "</html>\n"];
```

We match on the top element and embed the inner parts in an HTML body. Then we extract the string values of all motorcycle brands, sort them and removes duplicates by `remove_duplicates(value_of(select("bike/name/manufacturer", E)))`. We also process the substructure of the top element and pass it to a function that sorts all motorcycle information by brand according to the task formulation in the beginning of this example.

The next template matches on the `bike` element:

```
template(E = #xmlElement{name='bike'}) ->
  {value_of(select("name/manufacturer",E)), ["<dt>",xslapply(fun template/1,select("name",E)), "</dt>"],
  "<dd><ul>\n",
  "<li style=\"color:green\">Manufacturing year: ",xslapply(fun template/1,select("@year",E)), "</li>",
  "<li style=\"color:red\">Color: ",xslapply(fun template/1,select("@color",E)), "</li>\n",
  "<li style=\"color:blue\">Shape : ",xslapply(fun template/1,select("@condition",E)), "</li>\n",
  "</ul></dd>\n"]];
```

This creates a tuple with the brand of the motorcycle and the output format. We use the brand name only for sorting purpose. We have to end the template function with the “built in clause” `template(E) -> built_in_rules(fun template/1, E)`.

The entire program is `motorcycles2html.erl`<sup>18</sup>. If we run it like this `motorcycles2html:process_to_file('result_xs.html', 'motorcycles2.xml')`. The input is `motorcycles2.xml`<sup>19</sup> and you can check the result in the output file `result_xs.html`<sup>20</sup>.

---

<sup>16</sup>URL: <http://www.w3.org/TR/xpath>

<sup>17</sup>URL: [xmerl\\_xs\\_examples.html](http://xmerl_xs_examples.html)

<sup>18</sup>URL: [motorcycles2html.erl](http://motorcycles2html.erl)

<sup>19</sup>URL: [motorcycles2.txt](http://motorcycles2.txt)

<sup>20</sup>URL: [result\\_xs.html](http://result_xs.html)

## 1.2 xmerl Release Notes

This document describes the changes made to the xmerl application.

### 1.2.1 xmerl 1.0

#### Improvements and New Features

- The OTP release of xmerl 1.0 is mainly the same as xmerl-0.20 of <http://sowap.sourceforge.net/>. It is capable of parsing XML 1.0. There have only been minor improvements: Some bugs that caused an unexpected crash when parsing bad XML. Failure report that also tells which file that caused an error.  
Own Id: OTP-5174

#### Known Problems

- xmerl cannot fetch external entities that are referenced by an URL. Only references to the local file system are working.  
Own Id: OTP-5173
- xmerl cannot handle markup names that are Unicode encoded characters bigger than 256.  
Own Id: OTP-5172

There are also release notes for older versions<sup>21</sup>.

---

<sup>21</sup>URL: [notes\\_history.html](#)



# xmerl Reference Manual

## Short Summaries

- Erlang Module **xmerl** [page 14] – Functions for exporting XML data to an external format.
- Erlang Module **xmerl\_eventp** [page 17] – Simple event-based front-ends to `xmerl_scan` for processing of XML documents in streams and for parsing in SAX style.
- Erlang Module **xmerl\_scan** [page 18] – This module is the interface to the XML parser, it handles XML 1.0.
- Erlang Module **xmerl\_xpath** [page 21] – The `xmerl_xpath` module handles the entire XPath 1.0 spec XPath expressions typically occurs in XML attributes and are used to address parts of an XML document.
- Erlang Module **xmerl\_xs** [page 23] – Erlang has similarities to XSLT since both languages have a functional programming approach.

## xmerl

The following functions are exported:

- `callbacks(M::atom()) -> [atom()]`  
[page 14] Find the list of inherited callback modules for a given module.
- `export(Data::Content, Callback) -> ExportedFormat`  
[page 14] Equivalent to `export(Data, Callback, [])`.
- `export(Data::Content, Callback, RootAttrs::RootAttributes) -> ExportedFormat`  
[page 14] Exports normal, well-formed XML content, using the specified callback-module.
- `export_content(Es::Content, CBs::Callbacks) -> term()`  
[page 15] Exports normal XML content directly, without further context.
- `export_element(E, CB) -> term()`  
[page 15] Exports a normal XML element directly, without further context.
- `export_element(E, CB::CBs, CBstate::UserState) -> ExportedFormat`  
[page 15] For on-the-fly exporting during parsing (SAX style) of the XML document.
- `export_simple(Data::Content, Callback) -> ExportedFormat`  
[page 15] Equivalent to `export_simple(Data, Callback, [])`.

- `export_simple(Data::Content, Callback, RootAttrs::RootAttributes) -> ExportedFormat`  
[page 15] Exports "simple-form" XML content, using the specified callback-module.
- `export_simple_content(Data, Callback) -> term()`  
[page 16] Exports simple XML content directly, without further context.
- `export_simple_element(Data, Callback) -> term()`  
[page 16] Exports a simple XML element directly, without further context.

## xmerl\_eventp

The following functions are exported:

- `file_sax(Fname::string(), CallbackModule::atom(), UserState, Options::option_list()) -> NewUserState`  
[page 17] Parse file containing an XML document, SAX style.
- `stream(Fname::string(), Options::option_list()) -> xmlElement()`  
[page 17] Parse file containing an XML document as a stream, DOM style.
- `stream_sax(Fname, Callback::CallbackModule, UserState, Options) -> xmlElement()`  
[page 17] Parse file containing an XML document as a stream, SAX style.
- `string_sax(String::list(), CallbackModule::atom(), UserState, Options::option_list()) -> xmlElement()`  
[page 17] Parse file containing an XML document, SAX style.

## xmerl\_scan

The following functions are exported:

- `accumulate_whitespace(T::string(), S::global_state(), X3::atom(), Acc::string()) -> {Acc, T1, S1}`  
[page 19] Function to accumulate and normalize whitespace.
- `cont_state(S::global_state()) -> global_state()`  
[page 19] Equivalent to `cont_state(ContinuationState, S)`.
- `cont_state(X::ContinuationState, S::global_state()) -> global_state()`  
[page 19] For controlling the ContinuationState, to be used in a continuation function, and called when the parser encounters the end of the byte stream.
- `event_state(S::global_state()) -> global_state()`  
[page 19] Equivalent to `event_state(EventState, S)`.
- `event_state(X::EventState, S::global_state()) -> global_state()`  
[page 19] For controlling the EventState, to be used in an event function, and called at the beginning and at the end of a parsed entity.
- `fetch_state(S::global_state()) -> global_state()`  
[page 19] Equivalent to `fetch_state(FetchState, S)`.
- `fetch_state(X::FetchState, S::global_state()) -> global_state()`  
[page 19] For controlling the FetchState, to be used in a fetch function, and called when the parser fetch an external resource (eg.

- `file(Filename::string()) -> {xmlElement(), Rest}`  
[page 19] Equivalent to `file(Filename, [])`.
- `file(Filename::string(), Options::option_list()) -> {xmlElement(), Rest}`  
[page 19] Parse file containing an XML document.
- `hook_state(S::global_state()) -> global_state()`  
[page 20] Equivalent to `hook_state(HookState, S)`.
- `hook_state(X::HookState, S::global_state()) -> global_state()`  
[page 20] For controlling the HookState, to be used in a hook function, and called when the parser has parsed a complete entity.
- `rules_state(S::global_state()) -> global_state()`  
[page 20] Equivalent to `rules_state(RulesState, S)`.
- `rules_state(X::RulesState, S::global_state()) -> global_state()`  
[page 20] For controlling the RulesState, to be used in a rules function, and called when the parser store scanner information in a rules database.
- `string(Text::list()) -> {xmlElement(), Rest}`  
[page 20] Equivalent to `string(Test, [])`.
- `string(Text::list(), Options::option_list()) -> {xmlElement(), Rest}`  
[page 20] Parse string containing an XML document.
- `user_state(S::global_state()) -> global_state()`  
[page 20] Equivalent to `user_state(UserState, S)`.
- `user_state(X::UserState, S::global_state()) -> global_state()`  
[page 20] For controlling the UserState, to be used in a user function.

## xmerl\_xpath

The following functions are exported:

- `string(Str, Doc) -> docEntity()`  
[page 21] Equivalent to `string(Str, Doc, [])`.
- `string(Str, Doc, Options) -> docEntity()`  
[page 21] Equivalent to `string(Str, Doc, [], Doc, Options)`.
- `string(Str, Node, Parents, Doc, Options) -> docEntity()`  
[page 22] Extracts the nodes from the parsed XML tree according to XPath.

## xmerl\_xs

The following functions are exported:

- `built_in_rules(Fun, E) -> List`  
[page 23] The default fallback behaviour.
- `select(String::string(), E) -> E`  
[page 23] Extracts the nodes from the xml tree according to XPath.
- `value_of(E) -> List`  
[page 23] Concatenates all text nodes within the tree.
- `xslapply(Fun::Function, EList::list()) -> List`  
[page 24] `xslapply` is a wrapper to make things look similar to `xsl:apply-templates`.

# xmerl

Erlang Module

Functions for exporting XML data to an external format.

## Exports

`callbacks(M::atom()) -> [atom()]`

Find the list of inherited callback modules for a given module.

`export(Data::Content, Callback) -> ExportedFormat`

Equivalent to `export(Data, Callback, [])` [page 14].

`export(Data::Content, Callback, RootAttrs::RootAttributes) -> ExportedFormat`

Types:

- Content = [Element]
- Callback = atom()
- RootAttributes = [XmlAttribute]

Exports normal, well-formed XML content, using the specified callback-module.

Element is any of:

- #xmlText{}
- #xmlElement{}
- #xmlPI{}
- #xmlComment{}
- #xmlDecl{}

(See `xmerl.hrl` for the record definitions.) Text in `#xmlText{}` elements can be deep lists of characters and/or binaries.

`RootAttributes` is a list of `#XmlAttribute{}` attributes for the `#root#` element, which implicitly becomes the parent of the given `Content`. The tag-handler function for `#root#` is thus called with the complete exported data of `Content`. Root attributes can be used to specify e.g. encoding or other metadata of an XML or HTML document.

The `Callback` module should contain hook functions for all tags present in the data structure. A hook function must have the following format:

```
Tag(Data, Attributes, Parents, E)
```

where `E` is the corresponding `#xmlElement{}`, `Data` is the already-exported contents of `E` and `Attributes` is the list of `#xmlAttribute{}` records of `E`. Finally, `Parents` is the list of parent nodes of `E`, on the form `[{ParentTag::atom(), ParentPosition::integer()}]`.

The hook function should return either the data to be exported, or a tuple `{'#xml-alias#', NewTag::atom()}`, or a tuple `{'#xml-redefine#', Content}`, where `Content` is a content list (which can be on simple-form; see `export_simple/2` for details).

A callback module can inherit definitions from other callback modules, through the required function `'#xml-inheritance#' -> [ModuleName::atom()]`.

See also: `export/2` [page 14], `export_simple/3` [page 15].

```
export_content(Es::Content, CBs::Callbacks) -> term()
```

Types:

- Content = [Element]
- Callback = [atom()]

Exports normal XML content directly, without further context.

```
export_element(E, CB) -> term()
```

Exports a normal XML element directly, without further context.

```
export_element(E, CB::CBs, CBstate::UserState) -> ExportedFormat
```

For on-the-fly exporting during parsing (SAX style) of the XML document.

```
export_simple(Data::Content, Callback) -> ExportedFormat
```

Equivalent to `export_simple(Data, Callback, [])` [page 15].

```
export_simple(Data::Content, Callback, RootAttrs::RootAttributes) -> ExportedFormat
```

Types:

- Content = [Element]
- Callback = atom()
- RootAttributes = [XmlAttribute]

Exports “simple-form” XML content, using the specified callback-module.

Element is any of:

- {Tag, Attributes, Content}
- {Tag, Content}
- Tag
- IOString
- #xmlText{}
- #xmlElement{}
- #xmlPI{}
- #xmlComment{}
- #xmlDecl{}

where

- `Tag = atom()`
- `Attributes = [{Name, Value}]`
- `Name = atom()`
- `Value = IOString | atom() | integer()`

Normal-form XML elements can thus be included in the simple-form representation. Note that content lists must be flat. An `IOString` is a (possibly deep) list of characters and/or binaries.

`RootAttributes` is a list of:

- `XmlAttributes = #xmlAttribute{}`

See `export/3` for details on the callback module and the root attributes. The XML-data is always converted to normal form before being passed to the callback module.

*See also:* `export/3` [page 14], `export_simple/2` [page 15].

`export_simple_content(Data, Callback) -> term()`

Exports simple XML content directly, without further context.

`export_simple_element(Data, Callback) -> term()`

Exports a simple XML element directly, without further context.

# xmerl\_eventp

Erlang Module

Simple event-based front-ends to `xmerl_scan` for processing of XML documents in streams and for parsing in SAX style. Each contain more elaborate settings of `xmerl_scan` that makes usage of the customization functions.

## Exports

```
file_sax(Fname::string(), CallbackModule::atom(), UserState, Options::option_list())  
-> NewUserState
```

Parse file containing an XML document, SAX style. Wrapper for a call to the XML parser `xmerl_scan` with a `hook_fun` for using `xmerl` export functionality directly after an entity is parsed.

```
stream(Fname::string(), Options::option_list()) -> xmlElement()
```

Parse file containing an XML document as a stream, DOM style. Wrapper for a call to the XML parser `xmerl_scan` with a `continuation_fun` for handling streams of XML data. Note that the `continuation_fun`, `acc_fun`, `fetch_fun`, `rules` and `close_fun` options cannot be user defined using this parser.

```
stream_sax(Fname, Callback::CallbackModule, UserState, Options) -> xmlElement()
```

Types:

- `Fname` = `string()`
- `CallbackModule` = `atom()`
- `Options` = `option_list()`

Parse file containing an XML document as a stream, SAX style. Wrapper for a call to the XML parser `xmerl_scan` with a `continuation_fun` for handling streams of XML data. Note that the `continuation_fun`, `acc_fun`, `fetch_fun`, `rules`, `hook_fun`, `close_fun` and `user_state` options cannot be user defined using this parser.

```
string_sax(String::list(), CallbackModule::atom(), UserState, Options::option_list())  
-> xmlElement()
```

Parse file containing an XML document, SAX style. Wrapper for a call to the XML parser `xmerl_scan` with a `hook_fun` for using `xmerl` export functionality directly after an entity is parsed.

# xmerl\_scan

Erlang Module

This module is the interface to the XML parser, it handles XML 1.0. The XML parser is activated through `xmerl_scan:string/[1,2]` or `xmerl_scan:file/[1,2]`. It returns records of the type defined in `xmerl.hrl`. See also tutorial [page ??] on customization functions.

## Data Types

**abstract datatype:** `global_state()` The global state of the scanner, represented by the `#xmerl_scanner{}` record.

**abstract datatype:** `option_list()` Options allow to customize the behaviour of the scanner. See also tutorial [page ??] on customization functions.

Possible options are:

- `{acc_fun, Fun}` Call back function to accumulate contents of entity.
- `{continuation_fun, Fun} | {continuation_fun, Fun, ContinuationState}`  
Call back function to decide what to do if the scanner runs into EOF before the document is complete.
- `{event_fun, Fun} | {event_fun, Fun, EventState}` Call back function to handle scanner events.
- `{fetch_fun, Fun} | {fetch_fun, Fun, FetchState}` Call back function to fetch an external resource.
- `{hook_fun, Fun} | {hook_fun, Fun, HookState}` Call back function to process the document entities once identified.
- `{close_fun, Fun}` Called when document has been completely parsed.
- `{rules, ReadFun, WriteFun, RulesState} | {rules, Rules}` Handles storing of scanner information when parsing.
- `{user_state, UserState}` Global state variable accessible from all customization functions
- `{fetch_path, PathList}` PathList is a list of directories to search when fetching files. If the file in question is not in the `fetch_path`, the URI will be used as a file name.
- `{space, Flag}` 'preserve' (default) to preserve spaces, 'normalize' to accumulate consecutive whitespace and replace it with one space.
- `{line, Line}` To specify starting line for scanning in document which contains fragments of XML.
- `{namespace_conformant, Flag}` Controls whether to behave as a namespace conformant XML parser, 'false' (default) to not otherwise 'true'.
- `{validation, Flag}` Controls whether to process as a validating XML parser, 'false' (default) to not otherwise 'true'.
- `{quiet, Flag}` Set to 'true' if xmerl should behave quietly and not output any information to standard output (default 'false').

`{doctype_DTD, DTD}` Allows to specify DTD name when it isn't available in the XML document.

`{xmlbase, Dir}` XML Base directory. If using `string/1` default is current directory. If using `file/1` default is directory of given file.

`{encoding, Enc}` Set default character set used (default `utf-8`). This character set is used only if not explicitly given by the XML declaration. Note: If `ucs` is not available this **MUST** be set to a character set that is a true subset of Unicode and where each character only require a single byte. E.g., `iso-8859-1` or `equivalent`

## Exports

```
accumulate_whitespace(T::string(), S::global_state(), X3::atom(), Acc::string()) ->
  {Acc, T1, S1}
```

Function to accumulate and normalize whitespace.

```
cont_state(S::global_state()) -> global_state()
```

Equivalent to `cont_state(ContinuationState, S)` [page 19].

```
cont_state(X::ContinuationState, S::global_state()) -> global_state()
```

For controlling the `ContinuationState`, to be used in a continuation function, and called when the parser encounters the end of the byte stream. See tutorial [page ??] on customization functions.

```
event_state(S::global_state()) -> global_state()
```

Equivalent to `event_state(EventState, S)` [page 19].

```
event_state(X::EventState, S::global_state()) -> global_state()
```

For controlling the `EventState`, to be used in an event function, and called at the beginning and at the end of a parsed entity. See tutorial [page ??] on customization functions.

```
fetch_state(S::global_state()) -> global_state()
```

Equivalent to `fetch_state(FetchState, S)` [page 19].

```
fetch_state(X::FetchState, S::global_state()) -> global_state()
```

For controlling the `FetchState`, to be used in a fetch function, and called when the parser fetch an external resource (eg. a DTD). See tutorial [page ??] on customization functions.

```
file(Filename::string()) -> {xmlElement(), Rest}
```

Types:

- `Rest = list()`

Equivalent to `file(Filename, [])` [page 20].

```
file(Filename::string(), Options::option_list()) -> {xmlElement(), Rest}
```

Types:

- Rest = list()

Parse file containing an XML document

```
hook_state(S::global_state()) -> global_state()
```

Equivalent to `hook_state(HookState, S)` [page 20].

```
hook_state(X::HookState, S::global_state()) -> global_state()
```

For controlling the HookState, to be used in a hook function, and called when the parser has parsed a complete entity. See tutorial [page ??] on customization functions.

```
rules_state(S::global_state()) -> global_state()
```

Equivalent to `rules_state(RulesState, S)` [page 20].

```
rules_state(X::RulesState, S::global_state()) -> global_state()
```

For controlling the RulesState, to be used in a rules function, and called when the parser store scanner information in a rules database. See tutorial [page ??] on customization functions.

```
string(Text::list()) -> {xmlElement(), Rest}
```

Types:

- Rest = list()

Equivalent to `string(Test, [])` [page 20].

```
string(Text::list(), Options::option_list()) -> {xmlElement(), Rest}
```

Types:

- Rest = list()

Parse string containing an XML document

```
user_state(S::global_state()) -> global_state()
```

Equivalent to `user_state(UserState, S)` [page 20].

```
user_state(X::UserState, S::global_state()) -> global_state()
```

For controlling the UserState, to be used in a user function. See tutorial [page ??] on customization functions.

# xmerl\_xpath

Erlang Module

The `xmerl_xpath` module handles the entire XPath 1.0 spec XPath expressions typically occurs in XML attributes and are used to address parts of an XML document. The grammar is defined in `xmerl_xpath_parse.yrl`. The core functions are defined in `xmerl_xpath_pred.erl`.

Some useful shell commands for debugging the XPath parser

```
c(xmerl_xpath_scan).
yecc:yecc("xmerl_xpath_parse.yrl", "xmerl_xpath_parse", true, []).
c(xmerl_xpath_parse).

xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("position() > -1")).
xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("5 * 6 div 2")).
xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("5 + 6 mod 2")).
xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("5 * 6")).
xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("----6")).
xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("parent::node()")).
xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("descendant-or-self::node()")).
xmerl_xpath_parse:parse(xmerl_xpath_scan:tokens("parent::processing-instruction('foo')")).
```

## Data Types

```
docEntity() = xmlElement() | xmlAttribute() | xmlText() | xmlPI() | xmlComment()
```

```
nodeEntity() = xmlElement() | xmlAttribute() | xmlText() | xmlPI() | xmlNamespace() |
```

**abstract datatype:** `option_list()` Options allows to customize the behaviour of the XPath scanner.

Possible options are:

```
{namespace, #xmlNamespace} Set namespace nodes, from XmlNamespace, in
xmlContext
```

```
{namespace, Nodes} Set namespace nodes in xmlContext.
```

## Exports

```
string(Str, Doc) -> docEntity()
```

Equivalent to `string(Str, Doc, [])` [page 22].

```
string(Str, Doc, Options) -> docEntity()
```

Equivalent to `string(Str, Doc, [], Doc, Options)` [page 22].

`string(Str, Node, Parents, Doc, Options) -> docEntity()`

Types:

- `Str = xpathString()`
- `Node = nodeEntity()`
- `Parents = parentList()`
- `Doc = nodeEntity()`
- `Options = option_list()`

Extracts the nodes from the parsed XML tree according to XPath.

# xmerl\_xs

## Erlang Module

Erlang has similarities to XSLT since both languages have a functional programming approach. Using `xmerl_xpath` it is possible to write XSLT like transforms in Erlang.

XSLT stylesheets are often used when transforming XML documents, to other XML documents or (X)HTML for presentation. There are a number of brick-sized books written on the topic. XSLT contains quite many functions and learning them all may take some effort, which could be a reason why the author only has reached a basic level of understanding. This document assumes a basic level of understanding of XSLT.

Since XSLT is based on a functional programming approach with pattern matching and recursion it is possible to write similar style sheets in Erlang. At least for basic transforms. This document describes how to use the XPath implementation together with Erlangs pattern matching and a couple of functions to write XSLT like transforms.

This approach is probably easier for an Erlanger but if you need to use real XSLT stylesheets in order to “comply to the standard” there is an adapter available to the Sablotron XSLT package which is written in C++. See also the Tutorial [page ??].

## Exports

`built_in_rules(Fun, E) -> List`

The default fallback behaviour. Template funs should end with:

`template(E) -> built_in_rules(fun template/1, E).`

`select(String::string(), E) -> E`

Extracts the nodes from the xml tree according to XPath.

*See also:* `value_of/1` [page 23].

`value_of(E) -> List`

Types:

- `E = unknown()`

Concatenates all text nodes within the tree.

Example:

```
<xsl:template match="title">
  <div align="center">
    <h1><xsl:value-of select="." /></h1>
  </div>
</xsl:template>
```

becomes:

```
template(E = #xmlElement{name='title'}) ->
  ["<div align="center"><h1>",
   value_of(select(".", E)), "</h1></div>"]
```

```
xslapply(Fun::Function, EList::list()) -> List
```

Types:

- Function = () -> list()

xslapply is a wrapper to make things look similar to xsl:apply-templates.

Example, original XSLT:

```
<xsl:template match="doc/title">
  <h1>
    <xsl:apply-templates/>
  </h1>
</xsl:template>
```

becomes in Erlang:

```
template(E = #xmlElement{ parents=[{'doc',_|_}, name='title']) ->
  ["<h1>",
   xslapply(fun template/1, E),
   "</h1>"];
```

# Index of Modules and Functions

Modules are typed in *this* way.

Functions are typed in *this* way.

accumulate\_whitespace/1  
    xmerl\_scan , 19

built\_in\_rules/2  
    xmerl\_xs , 23

callbacks/1  
    xmerl , 14

cont\_state/1  
    xmerl\_scan , 19

cont\_state/2  
    xmerl\_scan , 19

event\_state/1  
    xmerl\_scan , 19

event\_state/2  
    xmerl\_scan , 19

export/2  
    xmerl , 14

export/3  
    xmerl , 14

export\_content/2  
    xmerl , 15

export\_element/2  
    xmerl , 15

export\_element/3  
    xmerl , 15

export\_simple/2  
    xmerl , 15

export\_simple/3  
    xmerl , 15

export\_simple\_content/2  
    xmerl , 16

export\_simple\_element/2  
    xmerl , 16

fetch\_state/1  
    xmerl\_scan , 19

fetch\_state/2  
    xmerl\_scan , 19

file/1  
    xmerl\_scan , 19, 20

file\_sax/1  
    xmerl\_eventp , 17

hook\_state/1  
    xmerl\_scan , 20

hook\_state/2  
    xmerl\_scan , 20

rules\_state/1  
    xmerl\_scan , 20

rules\_state/2  
    xmerl\_scan , 20

select/1  
    xmerl\_xs , 23

stream/1  
    xmerl\_eventp , 17

stream\_sax/4  
    xmerl\_eventp , 17

string/1  
    xmerl\_scan , 20

string/2  
    xmerl\_xpath , 21

string/3  
    xmerl\_xpath , 21

string/5  
    xmerl\_xpath , 22

- string\_sax/1
  - xmerl\_eventp* , 17
- user\_state/1
  - xmerl\_scan* , 20
- user\_state/2
  - xmerl\_scan* , 20
- value\_of/1
  - xmerl\_xs* , 23
- xmerl*
  - callbacks/1, 14
  - export/2, 14
  - export/3, 14
  - export\_content/2, 15
  - export\_element/2, 15
  - export\_element/3, 15
  - export\_simple/2, 15
  - export\_simple/3, 15
  - export\_simple\_content/2, 16
  - export\_simple\_element/2, 16
- xmerl\_eventp*
  - file\_sax/1, 17
  - stream/1, 17
  - stream\_sax/4, 17
  - string\_sax/1, 17
- xmerl\_scan*
  - accumulate\_whitespace/1, 19
  - cont\_state/1, 19
  - cont\_state/2, 19
  - event\_state/1, 19
  - event\_state/2, 19
  - fetch\_state/1, 19
  - fetch\_state/2, 19
  - file/1, 19, 20
  - hook\_state/1, 20
  - hook\_state/2, 20
  - rules\_state/1, 20
  - rules\_state/2, 20
  - string/1, 20
  - user\_state/1, 20
  - user\_state/2, 20
- xmerl\_xpath*
  - string/2, 21
  - string/3, 21
  - string/5, 22
- xmerl\_xs*
  - built\_in\_rules/2, 23
  - select/1, 23
- value\_of/1, 23
  - xslapply/2, 24
- xslapply/2
  - xmerl\_xs* , 24